

```
> restart; with(linalg);
```

Warning, the protected names norm and trace have been redefined and unprotected

```
[BlockDiagonal, GramSchmidt, JordanBlock, LUdecomp, QRdecomp, Wronskian, addcol,
  addrow, adj, adjoint, angle, augment, backsub, band, basis, bezout, blockmatrix, charmat,
  charpoly, cholesky, col, coldim, colspace, colspan, companion, concat, cond, copyinto, crossprod,
  curl, definite, delcols, delrows, det, diag, diverge, dotprod, eigenvals, eigenvalues, eigenvectors,
  eigenvects, entermatrix, equal, exponential, extend, ffgausselim, fibonacci, forwardsub, frobenius,
  gausselim, gaussjord, geneqns, genmatrix, grad, hadamard, hermite, hessian, hilbert, htranspose,
  ihermite, indexfunc, innerprod, intbasis, inverse, ismith, issimilar, iszero, jacobian, jordan, kernel,
  laplacian, leastsqrs, linsolve, matadd, matrix, minor, minpoly, mulcol, mulrow, multiply, norm,
  normalize, nullspace, orthog, permanent, pivot, potential, randmatrix, randvector, rank, ratform,
  row, rowdim, rowspace, rowspan, rref, scalarmul, singularvals, smith, stackmatrix, submatrix,
  subvector, subbasis, swapcol, swaprow, sylvester, toeplitz, trace, transpose, vandermonde,
  vecpotent, vectdim, vector, wronskian]
```

Problem 1

a) If you are given a stream function it exists, if it exists continuity must be satisfied. You can of course check it by finding velocity components and taking divergence of velocity and we give credit for either approach (5pts).

```
> psi:=y+4*x*y;
```

```
psi := y + 4 x y
```

b) finding velocity components (5 pts)

```
> u:=diff(psi,y); v:=-diff(psi,x); contcheck:=(diff(u,x)+diff(v,y));
```

```
u := 1 + 4 x
```

```
v := -4 y
```

```
contcheck := 0
```

To check if flow is irrotational we integrate u with respect to x and v with respect to y, recognizing that since they are partial derivatives we will get undetermined functions of y and x respectively viz.

```
> phi1:=int(u,x)+fy;
```

```
phi1 := x + 2 x^2 + fy
```

```
> phi2:=int(v,y)+fx;
```

```
phi2 := -2 y^2 + fx
```

Reconciling phi1 and phi2 it is clear that fy must be  $-2 y^2$  and fx must be  $x+2x^2$ .

```
> phi:=x+2*x^2-2*y^2;
```

```
phi := x + 2 x^2 - 2 y^2
```

5pts for knowing you had to integrate each velocity, 5 pts for properly reconciling the result

5 pts for showing the potential exists (either by the above result or by checking that the vorticity was zero viz

```
> vort:=diff(v,x)-diff(u,y);
```

```
vort := 0
```

partd) To compute volume flow rate there are 3 ways 1) difference in value of psi passing through the two points at the end of the lines 2) draw a control volume in the shape of the right triangle that the given edge creates (I showed this in class Friday and the math details are below) or 3) evaluate the double integral of vector V dot vector capital dS (for surface) over the surface created by the line from (0,0) to (1,2). I will do them in reverse order. First, method 3. Recall that vector capital dS is vector normal to the surface times the area (scalar) element. This area element is reduced to a line element ds since we need the volume flow rate per unit depth. This means we will, later, need to express the integrand in terms of s. Before doing this, lets find the integrand by first finding the normal to the edge (or surface if we imagine the edge extending into the board).

```
> edge:=[1,2]; nedge:=[-edge[2],edge[1]]/norm(edge,2);
```

```
edge := [1, 2]
```

```
nedge :=  $\frac{1}{5}[-2, 1]\sqrt{5}$ 
```

Since we want to find the flow rate through the edge with this normal we r

```
> v_vect:=[u,v]; vdotn:=dotprod(v_vect,nedge);
```

```
v_vect := [1 + 4 x, -4 y]
```

```
vdotn :=  $-\frac{2(1+4x)\sqrt{5}}{5} - \frac{4y\sqrt{5}}{5}$ 
```

note that this is a function of x and y and we want to get it in terms of s and then integrate

```
> vdotn_of_s:=simplify(subs({x=s/sqrt(5),y=2*s/sqrt(5)}, vdotn));
volflow:=int(vdotn_of_s,s=0..sqrt(5));
```

```
vdotn_of_s :=  $-\frac{2\sqrt{5}}{5} - \frac{16s}{5}$ 
```

```
volflow := -10
```

Scoring: normal=1pt, vdotn in terms of x 1pt, vdotn in terms of s 2 pts, integral setup and evaluated correctly 1 pt

If you really wanted to avoid the tricky substitution of s into x and y in part d you COULD have drawn a control volume in the shape of a triangle (e.g. a line from (0,0) to (1,0) and another line from (1,0) to (1,2) and then computed the flow rate through those two surfaces:

```
> VolFlowRateOther:=int(subs(y=0,v),x=0..1)+int(subs(x=1,u),y=0..2);
```

```
VolFlowRateOther := 10
```

and then noted that by conservation of mass the volume flow rate into the surface requested must be the negation of the Volume\_Flow\_Rate\_Other. By far the easiest way is the method 1) described above

```
> psi_SL_12:=subs({x=1,y=2},psi); psi_SL_00:=subs({x=0,y=0},psi);
mass_flow_line00_12:=psi_SL_12-psi_SL_00;
```

```
psi_SL_12 := 10
```

```
psi_SL_00 := 0
```

```
mass_flow_line00_12 := 10
```

Part e. You could use the same approach as part e or you could use the fact that  $\mathbf{V} \cdot d\mathbf{s}$  is path independent for an irrotational flow viz

```
> intvdots:=int(subs(y=0,u),x=0..1)+int(subs(x=1,v),y=0..2);
dsvect:=edge/norm(edge,2);
vdotds:=dotprod(v_vect,dsvect);
integrand2:=subs({x=s/sqrt(5),y=2*s/sqrt(5)},vdotds);
intvdots2:=int(integrand2,s=0..sqrt(5));
```

$$\text{intvdots} := -5$$

$$dsvect := \frac{1}{5} [1, 2] \sqrt{5}$$

$$vdotds := \frac{(1+4x)\sqrt{5}}{5} - \frac{8y\sqrt{5}}{5}$$

$$\text{integrand2} := \frac{\left(1 + \frac{4s\sqrt{5}}{5}\right)\sqrt{5}}{5} - \frac{16s}{5}$$

$$\text{intvdots2} := -5$$

Note that you do not get the same thing on partd that is the line integral of  $\mathbf{V} \cdot d\mathbf{s}$  is NOT the same as the area integral  $\mathbf{V} \cdot d\mathbf{S}$  (vector signs on all V's and S's).

Problem 2.

given the psi, part a just requires us to differentiate psi appropriately to find the velocity components  $V_r$  and  $V_\theta$

```
> psi:=Vinf*r*sin(theta)*(1-R^2/r^2)+Gamma*ln(r)/2/Pi;Vr:=diff(psi,theta)/r;
Vtheta:=-diff(psi,r); simplify(subs(Gamma=0,%));
```

$$\psi := V_{inf} r \sin(\theta) \left(1 - \frac{R^2}{r^2}\right) + \frac{1}{2} \frac{\Gamma \ln(r)}{\pi}$$

$$V_r := V_{inf} \cos(\theta) \left(1 - \frac{R^2}{r^2}\right)$$

$$V_\theta := -V_{inf} \sin(\theta) \left(1 - \frac{R^2}{r^2}\right) - \frac{2 V_{inf} \sin(\theta) R^2}{r^2} - \frac{\Gamma}{2 r \pi} - \frac{V_{inf} \sin(\theta) (r^2 + R^2)}{r^2}$$

There was considerable stress on the exam about whether there should be  $\ln(r/R)$  instead of  $\ln(r)$  but if you reviewed your notes you will see this is just a constant that we added to make the psi = 0 streamline the body. It has no effect on this problem. Below I prove that by getting the same velocity components

```
> psi:=Vinf*r*sin(theta)*(1-R^2/r^2)+Gamma*ln(r/R)/2/Pi;Vr:=diff(psi,theta)/r;
Vtheta:=-diff(psi,r); simplify(subs(Gamma=0,%));
```

$$\psi := V_{inf} r \sin(\theta) \left( 1 - \frac{R^2}{r^2} \right) + \frac{1}{2} \frac{\Gamma \ln\left(\frac{r}{R}\right)}{\pi}$$

$$V_r := V_{inf} \cos(\theta) \left( 1 - \frac{R^2}{r^2} \right)$$

$$V_{\theta} := -V_{inf} \sin(\theta) \left( 1 - \frac{R^2}{r^2} \right) - \frac{2 V_{inf} \sin(\theta) R^2}{r^2} - \frac{\Gamma}{2 r \pi}$$

$$- \frac{V_{inf} \sin(\theta) (r^2 + R^2)}{r^2}$$

Note also, it is perfectly acceptable to me for you to use the formulas for the velocity components that we derived in class since you were not asked to compute them on this exam. If fact, if you set your integral up correctly below you see you only needed  $V_{\theta}$

for part a

We are going to be evaluating  $-V \cdot ds$  around a circle of radius  $2R$ . We have  $V$  (the velocity vector above) we need  $ds$ . It is obviously easiest to do this integral in polar coordinates so  $ds = r d\theta \hat{e}_{\theta}$  where  $\hat{e}_{\theta}$  is the unit vector in the  $\theta$  direction. Since we are going to take the dot product of  $ds$  and  $V$  it is clear that only the  $V_{\theta}$  component is going to be relevant and thus our integrand is

$-V_{\theta} r d\theta$ . The last step is to be sure that we get the limits correct. I hope it is obvious that to go around the circle requires  $\theta$  to go from  $0$  to  $2\pi$ .

```
> pta:=int(-Vtheta*r,theta=0..2*Pi);
```

$$pta := \Gamma$$

Which initially troubles us until we remember that it is only the vortex solution that has any circulation and it is the same at any radius. You get 2 points for drawing your figure correctly, labeling  $ds$ . 2 points for getting  $ds$  correct. 2 points for correct integrand, 2 pts for correct limit and 2 points for correct final answer. Some of you felt more comfortable substituting  $r=2R$  in earlier which should go like this:

```
> integrand:=simplify(subs(r=2*R,-Vtheta*r)); pta:=int(integrand,theta=0..2*Pi);
```

$$integrand := \frac{1}{2} \frac{5 V_{inf} \sin(\theta) R \pi + \Gamma}{\pi}$$

$$pta := \Gamma$$

part b

$L' = \rho V_{inf} \Gamma$  (WORTH 5 pts)

[ Problem 3

```
> restart;
```

[ Declare the variables specific to this problem. This time it is the coordinates of the edge intersections

```
> x1:=-4; y1:=0; x2:=0; y2:=3; x3:=4; y3:=0; x4:=0; y4:=-3;
```

$$x1 := -4$$

$$y1 := 0$$

$$x2 := 0$$

$$y2 := 3$$

$$x3 := 4$$

$$y3 := 0$$

$$x4 := 0$$

$$y4 := -3$$

For row 3, column 3, you have  $i=j$  and thus you do not use the  $L_{ij}$  formula rather you write down that this coefficient =  $1/2$  (WORTH 5 PTS).

For row 3, column 4 you have  $i=3, j=4$  and thus you have to set up the interpolation for integration over the 4th panel.

For segment 4, define its root edge coordinates  $(x_b, y_b)$ , and ending edge coordinates  $(x_e, y_e)$  and the length of the edge

```
> xb:=x4; yb:=y4; xe:=x1; ye:=y1;  
ledge:=sqrt((xb-xe)^2+(yb-ye)^2);
```

```
>
```

$$x_b := 0$$

$$y_b := -3$$

$$x_e := -4$$

$$y_e := 0$$

$$ledge := 5$$

With this information we can write the parameterization of the edge as a function of distance along the edge (WORTH 5 PTS)

```
> xjs:=xb+(xe-xb)*s/ledge; evalf(%);  
yjs:=yb+(ye-yb)*s/ledge;evalf(%);
```

$$xjs := -\frac{4s}{5}$$

$$-0.8000000000s$$

$$yjs := -3 + \frac{3s}{5}$$

$$-3. + 0.6000000000s$$

The integrand also needs the coordinates of the  $i$ th control point and the angle that the normal to this control point's edge makes with the  $x$  axis,  $\beta_{ai}$ . In this case the normal is more easily calculated using the 3-4-5 triangle (the following is a general formula to find the outward normal to any edge where you know the starting and ending point of the edge)

```
> sx:=x4-x3; sy:=y4-y3; tx:=sy/ledge; ty:=-sx/ledge;  
nx:=tx*`if`(sx*ty-tx*sy>0,1,-1); ny:=ty*`if`(sx*ty-tx*sy>0,1,-1);
```

$$s_x := -4$$

$$s_y := -3$$

$$t_x := \frac{-3}{5}$$

$$t_y := \frac{4}{5}$$

$$nx := \frac{3}{5}$$

$$ny := \frac{-4}{5}$$

cos(beta\_3)=nx, sin(beta\_3)=ny. (the above or any alternate form to compute the values of cos(beta\_3) and sin(beta\_3) WORTH 5 PTS).

Now plugging in the previous work to compute I\_{34} numerator and denominator

```
> xcpi:=(x4+x3)/2; ycpi:=(y4+y3)/2;
  inumer:=(xcpi-xjs)*nx+(ycpi-yjs)*ny;
  evalf(%); idenom:=simplify((xcpi-xjs)^2+(ycpi-yjs)^2); evalf(%);
```

$$xcpi := 2$$

$$ycpi := \frac{-3}{2}$$

$$inumer := \frac{24s}{25}$$

$$0.9600000000s$$

$$idenom := \frac{25}{4} + \frac{7}{5}s + s^2$$

$$6.250000000 + 1.400000000s + s^2$$

putting it all together (WORTH 5 pts)

```
> integrate(1/2/Pi*inumer/idenom,s=0..ledge);evalf(%);
```

$$\frac{1}{50} \frac{24 \ln(5) - 7 \arctan\left(\frac{7}{24}\right) - 24 \ln(3) - 12 \ln(17) + 7 \arctan\left(\frac{19}{8}\right)}{\pi}$$

$$0.09879948472$$

>

What changes for column 1 and column 2 entries for row 3? From the big picture, what changes is the panel you are integrating over. In detail, the length of the panel remains the same so only the parameterization of  $x_j$  and  $y_j$  (in terms of  $s$ ) change. Roughly 5 equation blocks back we defined  $x_b, y_b, x_e$  and  $y_e$ . These change such that: for column 1  $b=1, e=2$ , for column 2,  $b=2, e=3$ . If one re-evaluates the same formulas after that with only these changes, the correct results will be found. Note the  $x_p$ , the  $y_p$ , and the normals/beta angles stay the same because they correspond to the control point (i) which is the row and not changing. (WORTH 5 pts....to get full credit you only had to say what changed not what stayed the same....I add that just for completeness)